

Processamento de Dados aplicado à Geociências

AULA 8: Comandos de repetição Repita-até

UNIVERSIDADE FEDERAL DE PELOTAS
CENTRO DE DESENVOLVIMENTO TECNOLÓGICO
CURSO SUPERIOR DE TECNOLOGIA EM
GEOPROCESSAMENTO

Professor: Guilherme Tomaschewski Netto
guilherme.netto@inf.ufpel.edu.br



Repita-até

- A estrutura de repetição Repita-até é utilizada quando um conjunto de comandos deve ser executado repetidamente, até que uma condição (expressão lógica) seja verdadeira.
- Sintaxe:
Repita
 <lista_comandos>
até <condição>

Semântica

- A execução do comando ocorre da seguinte maneira:
 - O comando é executado uma vez.
 - Após, a condição é testada: se ela for falsa, a lista de comandos é executada novamente e este processo é repetido até que a condição seja verdadeira, quando então a execução prossegue pelo comando imediatamente seguinte ao final da construção.

Fluxograma





Diagrama de Chapin



Repita-até

- Solução para o problema do reajuste de 50 produtos usando a estrutura de repetição Repita-até.

Repita-até

Algoritmo

Declare c, v, tot, i **Numérico**

tot := 0

i := 1

Repita

Escreva “Informe o código e o valor”, \n

Leia c, v

Se v < 100

então v := v+(v*0.1)

senão v := v+(v*0.03)

fim_se

Escreva “Código”, c, “Valor”, v, \n

i := i + 1

tot := tot + v

até i > 50

Escreva “Valor total”, tot, \n

fim_algoritmo

Repita-até x Enquanto-faça

- A estrutura de repetição Repita-até difere da estrutura Enquanto-faça pelo fato de a lista de comandos ser executada uma ou mais vezes (pelo menos uma vez), ao passo que na estrutura Enquanto-faça a lista de comandos é executada zero ou mais vezes (possivelmente nenhuma vez).
- Isto acontece porque na estrutura Repita-até o teste é feito no final da construção, ao contrário do que acontece na estrutura Enquanto-faça, onde o teste da condição é efetuado no início da estrutura.

Exemplo

- Algoritmo que escreva os números de 1 à 10

```
Algoritmo
  Declare i Numérico
  i:=1
  Repita
    Escreva i,\n
    i:=i+1
  até i>10
fim_algoritmo
```

```
Algoritmo
  Declare i Numérico
  i:=0
  Repita
    i:=i+1
    Escreva i,\n
  até i=10
fim_algoritmo
```

```
Algoritmo
  Declare i Numérico
  i:=1
  Enquanto i<=10 faça
    Escreva i,\n
    i:=i+1
  fim_enquanto
fim_algoritmo
```

Repita-até x Enquanto-faça

- A estrutura Repita-até também difere da estrutura Enquanto-faça no que se refere à inicialização da variável de controle, visto que na estrutura Repita-até a variável pode ser inicializada ou lida dentro do laço.

Exemplo

- Escreva um algoritmo que lê valores um número não determinado de vezes e escreve a raiz quadrada de cada um destes valores. O algoritmo deve parar quando um valor negativo for digitado.

Exemplo

Enquanto-faça

```

Algoritmo
  Declare Num Numérico
  Escreva "Digite um valor", \n
  Leia Num
  Enquanto Num >= 0 faça
    Escreva "Raiz de ", Num, " é", Num**(1/2), \n
    Escreva "Digite um valor", \n
    Leia Num
  fim_enquanto
fim_algoritmo

```

Repita-até

```

Algoritmo
  Declare Num Numérico
  Repita
    Escreva "Digite um valor", \n
    Leia Num
    Se Num >= 0
      então
        Escreva "Raiz de ", Num, " é", Num**(1/2), \n
    fim_se
  até num < 0
fim_algoritmo

```

Algoritmo de Fibonacci

- Faça um algoritmo que escreva os 6 primeiros números da série de Fibonacci utilizando a estrutura de repetição Repita-até.

Algoritmo de Fibonacci Com Repita-até

Algoritmo

Declare A,B, i Numérico

A:=0

B:=1

i:=1

Repita

 A:= A+B

 B:= A-B

 Escreva B, \n

 i:= i+1

até i > 6

Fim_Algoritmo

Teste de mesa

Linha	Algoritmo
1	Algoritmo
2	Declare A,B,i Numerico
3	A:=0
4	B:=1
5	i:= 1
6	Repita
7	A := A+B
8	B := A - B
9	Escreva B, \n
10	i:= i+1
11	até i>6
12	fim_algoritmo

Linha	Teste de mesa			
	A	B	i	Saída
2	?	?	?	
3	0	?	?	
4	0	1	?	
5	0	1	1	
6	0	1	1	
7	1	1	1	
8	1	0	1	
9	1	0	1	0
10	1	0	2	
11	1	0	2	
7	1	0	2	
8	1	1	2	
9	1	1	2	1
10	1	1	3	
11	1	1	3	
7	2	1	3	
8	2	1	3	
9	2	1	3	1
10	2	1	4	
11	2	1	4	

Teste de mesa

Continuação

Linha	Algoritmo
1	Algoritmo
2	Declare A,B,i Numerico
3	A:=0
4	B:=1
5	i:= 1
6	Repita
7	A := A+B
8	B := A - B
9	Escreva B, \n
10	i:= i+1
11	até i>6
12	fim_algoritmo

Linha	Teste de mesa			
	A	B	i	Saída
11	2	1	4	
7	3	1	4	
8	3	2	4	
9	3	2	4	2
10	3	2	5	
11	3	2	5	
7	5	2	5	
8	5	3	5	
9	5	3	5	3
10	5	3	6	
11	5	3	6	
7	8	3	6	
8	8	5	6	
9	8	5	6	5
10	8	5	7	
11	8	5	7	

Exercício 1

- Escreva um algoritmo que escreva a tabuada do 5 (de 1 à 10).
- Fazer o algoritmo utilizando a estrutura de repetição Repita-até

Exercício 1 - Solução

Algoritmo

Declare Num, Tabuada **Numérico**

Num:=1

Repita

Tabuada:= Num*5

Escreva 'Tabuada', Num, ' * 5:', Tabuada, \n

Num:=Num+1

até Num>10

fim_algoritmo

Exercício 2- Médias

- Faça um algoritmo que, para 10 alunos: leia o nome do aluno e duas notas deste aluno e que escreva o nome do aluno, informando se ele está aprovado caso sua média seja maior ou igual a 7, juntamente com a sua média e reprovado, caso contrário.
- Fazer o algoritmo utilizando a estrutura de repetição Repita-até

Exercício 2- Médias

Algoritmo

```
Declare Nota1, Nota2, Media, i Numerico
Declare Nome Literal
i:= 1
Repita
    Escreva 'Nome do aluno',\n
    Leia Nome
    Escreva 'Entre com a primeira nota',\n
    Leia Nota1
    Escreva 'Entre com a segunda nota',\n
    Leia Nota2
    Media := (Nota1+Nota2)/2
    Se Media >= 7
        então Escreva 'Aluno', Nome, 'Aprovado com nota', Media,\n
        senão Escreva 'Aluno', Nome, 'Reprovado com nota', Media,\n
    fim_se
    i:=i+1
até i>10
fim_algoritmo
```

Exercício 3

- Escreva um algoritmo para ler 2 valores e se o segundo valor informado for ZERO, deve ser mostrada uma mensagem de valor inválido e deve ser lido um novo valor (ou seja, para o segundo valor não pode ser aceito o valor zero), e imprimir o resultado da divisão do primeiro valor lido pelo segundo valor lido.
- Fazer o algoritmo utilizando a estrutura de repetição Repita-até

Exercício 3 - Solução

Algoritmo

```
Declare Valor1, Valor2, Divisao Numérico
Escreva 'Entre com o primeiro valor',\n
Leia Valor1
Repita
    Escreva 'Entre com o segundo valor',\n
    Leia Valor2
    Se Valor2=0
        então
            Escreva 'Zero é um valor inválido',\n
        fim_se
até Valor2<>0
Divisao:= Valor1/Valor2
Escreva 'A divisão de', Valor1, 'por', Valor2, 'é', Divisao,\n
fim_algoritmo
```

Exercício 4

- Faça um algoritmo que escreva os 100 primeiros números pares.
- Fazer o algoritmo utilizando a estrutura de repetição Repita-até

Exercício 4 - Solução

Algoritmo

Declare Par, Contador Numérico

Par:=0

Contador:=1

Repita

 Escreva Par,\n

 Par:=Par+2

 Contador:=Contador+1

até Contador>100

fim_algoritmo

Exercício 5

- Faça um algoritmo que leia um número M que indica quantos valores devem ser lidos a seguir. Para cada número lido, mostre o dobro deste valor.
- Fazer o algoritmo utilizando a estrutura de repetição Repita-até

Exercício 5 - Solução

Algoritmo

Declare M, Contador, Valor, Dobro **Numérico**

Escreva 'Digite a quantidades de valores a ler', \n

Leia M

Contador:=1

Repita

Escreva 'Digite o valor a calcular o dobro', \n

Leia Valor

Dobro:= Valor*2

Escreva 'O dobro de', Valor, 'é:', Dobro, \n

Contador := Contador +1

até Contador>M

fim_algoritmo

Exercício 6

- Faça um algoritmo que leia um número M que indica quantos valores devem ser lidos a seguir e mostre a média aritmética dos valores lidos.
- Fazer o algoritmo utilizando a estrutura de repetição Repita-até

Exercício 6 - Solução

Algoritmo

Declare M, Contador, Valor, Total, Media **Numérico**

Escreva 'Digite a quantidades de valores a ler', \n

Leia M

Contador:=1

Total :=0

Repita

Escreva 'Digite um valor', \n

Leia Valor

 Total:= Total+Valor

 Contador := Contador+1

até Contador>M

Media:= Total/M

Escreva 'A média dos valores lidos é:', Media

fim_algoritmo